
DoFiles/PUBLICUSE_00_Master_Analysis_v1.do
/*** I. Overview: PUBLIC RELEASE: File to set up any analysis and create the working dataset for the regressions
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com

****************************************
	*** II. Version History ****
****************************************
	2019/01/08, GS,    created from 00_Master_Analysis_v0
	2019/01/09, GS, v1 edited directories for Public Release

****************************************
	*** III. Inputs ****
****************************************
		Data for public release: "$data/01_indivdatawide_PandP.dta", clear

>>>>>	Add a new researcher after line 35 and a new directory to execute all commands. <<<<<<
		
****************************************
	*** IV. Outputs ****
****************************************
		All result presented in the AEA PandP Article "Parental Aspirations for Children's Education: Experimental Evidence from Ethiopia" 
		by Bernard Tanguy, Dercon Stefan, Orkin Kate, Seyoum Taffesse Alemayehu, 2019.
		
		
********************************************************************************/
*** Pre-Amble ****
********************************************************************************
clear all
clear mata
macro drop all
set maxvar 20000
set matsize 9000

	*Researcher
*local researcher="KO"  //uncomment if not needed
local researcher="GS"

if "`researcher'"=="KO" global desktop /Users/Kate/Dropbox/Research_projects
if "`researcher'"=="GS" global desktop 	"C:/Users/ball4189/Dropbox/aspirations_small_folder/PandP/01_Final"

********************************************************************************
*** 0. Directories ****
*******************************************************************************
global folder 	"$desktop/Orkin_08_Replication_Files"
global data 	"$folder/Data"
global dofiles 	"$folder/DoFiles"
global output 	"$folder/Logs"

********************************************************************************
*** 1. Define outcome variables and groups
********************************************************************************
local asp_educ		A_4_2  A_4_2_D
local educ_PandP2	kidsinsch715 time_sch_715_hh time_study_715_hh raw_schexpchild

local ind_sr	   	asp_educ 
local hh_sr	  	    educ_PandP2

********************************************************************************
*** 2. Data Robustness Switches **** (always check them)
********************************************************************************				
capture  log close
log		 using 				"$output/20190110" , replace			//can be commented, requires manual update of the file name.

**Used**
global 	has_controls 		"on"									//can be on or off. If on, then we add the condition that the respondent has to have all controls used in the regressions for column (2) and (4). 
global  time_with_zeroes 	"on"									//can be on or off. Lets you add zeroes to the time variables where there are no kids in the specific age-group, but only for the variables that were reconstructed for the specific age-groups.

**Not used**
global 	agegroup 			"715"									//can be 715, for now this local is added to add the new dummies and help keeping track on what outcomes we are focusing on. With some extra coding 620 can also be added back as an option.
global 	outliers4SD			"off"									//can be on or off. Lets you trim the time variables for any outlier with a value above/below 4 standard deviations of the mean.


********************************************************************************
*** 3. Define clustering and controls ****
********************************************************************************

global individual_controls 		male age regrade_ID single E_1_6 E_1_7			//For panel B we drop male, for panel C we drop regrade_ID
global ind_hh_controls 			$individual_controls hhsize__1 n_children__1
global r2_controls 				$ind_hh_controls  
global r1_controls 				$r2_controls	

di "$r2_controls"

global cluster " robust" /*" clustered at household level"*/

if "$has_controls"=="on" {
					local cond1 "has_all_controls_r1==0"		//adds a restriction on the sample at baseline to have all of the controls used in the regression
					}

					
********************************************************************************
*** 4. Opens the dataset ****
********************************************************************************
	///////////////////////////////////////////////////
		use "$data\01_indivdatawide_PandP.dta", clear
	///////////////////////////////////////////////////

********************************************************************************
*** 5. Sequence of Do-Files ****
********************************************************************************

include "$dofiles\PUBLICUSE_01_Variable_Construction_v1.do"					//Variables constructed for the analysis
include "$dofiles\PUBLICUSE_02_Analysis_Table_1_PanelA_v0.do"
include "$dofiles\PUBLICUSE_02_Analysis_Table_1_PanelB_v0.do"
include "$dofiles\PUBLICUSE_02_Analysis_Table_1_PanelC_v0.do"
include "$dofiles\PUBLICUSE_02_Analysis_Table_2_v0.do"

exit




DoFiles/PUBLICUSE_01_Variable_Construction_v1.do
/*** I. Overview: FOR PUBLIC RELEASE: File to create the final variables used in the P&P analysis.
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com

****************************************
*** II. Version History ****
****************************************
2019/01/08, GS, created from 01_Data_Preparation_v3
2019/01/10, GS, v1 update, adds labels and cleans a few unncessary commands

****************************************
		*** III. Inputs ****
****************************************
The individual-level dataset: 			"$path6/indivdatawide_PandP.dta"


********************************************************************************/


********************************************************************************
*** Construction of Outcomes for Table 1****
********************************************************************************
			local verb aspires 						//We are doing aspirations
			local variables A_4_2					//We are doing aspirations

			local n : word count `verb'					//Setting the counter for parallel list loop
			forvalues i=1/`n' {

			local v : word `i' of `variables'			//this is just a fancy way of getting the right label to the right variable.
			local l : word `i' of `verb'

				
					forvalues x=1/2{
							gen `v'_D__`x' = (`v'__`x'==13 | `v'__`x'==14 | `v'__`x'==15 | `v'__`x'==16)
							replace `v'_D__`x' = . if `v'__`x'==.
							label var `v'_D__`x' "\specialcell{[=1] if `l' child to go beyond secondary}"
									}
										}
										di "Closes values loop"
							
********************************************************************************
*** Constructing the dimensions of heterogeneity ****
********************************************************************************

		*Female child

						gen treatabovefemale_child__2=female_child__2*treatment  		//DOING interaction with the sex of the child in the Aspirations module - not the baseline.
						gen above_med_init_female_child__2=female_child__2
						label var treatabovefemale_child__2 	  "Treated and \\Female  Child r2"
						label var above_med_init_female_child__2  "Female Child r2"
	
		*Female			
						gen female__2=female							//just for coding convenience
						label var female__2 "=1 if respondent is female"

	
		*No education
							foreach x in regrade_ID {
							xtile med_init_`x'1 = `x', nq(2)
							gen above_med_init_`x' =0 if med_init_`x'1==1
							replace above_med_init_`x' =1 if med_init_`x'1==2
								}
							gen no_educ=1 if above_med_init_regrade_ID==0
							replace no_educ=0 if above_med_init_regrade_ID==1
						

		*Defining the difference for girls dimension of heterogeneity at the household level

			forvalues x=1/2{
				gen kidsinhh715_F__`x'  = (kidsinhh711_F__`x'+kidsinhh1215_F__`x')			//Creating total number of female kids aged 7-15 by round
							}
				
		*Generating the interaction to control for those that don't have kids, who enter the ITT regression

			forvalues x=1/2{
			foreach agegroup in 715 { 
				gen sexratio`agegroup'__`x'    =  kidsinhh`agegroup'_F__`x' /kidsinhh`agegroup'__`x'	//Creating ratio of female kids aged in specific age-group out of all kids in this age-group by round
				gen nokidsinhh`agegroup'__`x'=(kidsinhh`agegroup'__`x'==0)								//creating a dummy equal to one if the household doesn't have children in this agegroup
				label var nokidsinhh`agegroup'__`x' "=1 if hh doesn't have children in the `agegroup' age-group for round `x'"
				gen wm_sexratio`agegroup'__`x'=sexratio`agegroup'__`x'									//wm stands for with missing - because this variable will be...
				label var wm_sexratio`agegroup'__`x' "Share of girls in the `agegroup' age-group with missing obs replaced as zeroes"
				replace wm_sexratio`agegroup'__`x'=0 if  nokidsinhh`agegroup'__`x'==1					//...replacing the femaleshare to zero for those households with no children in this age-group
				}
				}
				
				
********************************************************************************
*** Imputations on outcome variables****
********************************************************************************

	*Trimming of values above 4SD away from the mean
	if "$outliers4SD" == "on" {
				foreach round in 1 2 {
				foreach agegroup in ${agegroup}{
				foreach x in time_study_`agegroup'_hh__`round'  time_sch_${agegroup}_hh__`round'{	
					bysort round: egen mean_`x'=mean(`x') 
					bysort round: egen sd_`x'=sd(`x') 
					bysort round: gen index_`x'=(`x'-mean_`x')/sd_`x' if `x'~=.	

					gen outlier_`x'=1 if index_`x'>$outlierSD | index_`x'<-$outlierSD
					replace outlier_`x'=0 if index_`x'<=$outlierSD  & index_`x'>=-$outlierSD 
					replace  outlier_`x'=. if index_`x'==. 
					drop mean_`x' sd_`x' index_`x'

				replace `x'=. if outlier_`x'==1
						}
						di "close x bracket"
						}
						di "close round bracket"
						}
						di "close agegroup bracket"
						}
						di "close outliers bracket"

	*Imputing a value of zero for time variables to not drop any households who enter the ITT regression
	if "$time_with_zeroes"=="on" {				
				forvalues x=1/2{																	
				foreach agegroup in 715 { 															
						replace time_study_`agegroup'_hh__`x'=0 if nokidsinhh`agegroup'__`x'==1  & time_study_`agegroup'_hh__`x'==.
						replace time_sch_`agegroup'_hh__`x'=0 if nokidsinhh`agegroup'__`x'==1  & time_sch_`agegroup'_hh__`x'==.	
							}
							di "close agegroup bracket"
							}
							di "close x bracket"
							}
							di "close zeroes bracket"
							
	*Making sure that col(1) and col (2) have the same number of obs, and that col(3) and col(4) also have the same numbers of obs.
						foreach x in 1 {
			gen has_all_controls_r`x'=0
			label var has_all_controls_r`x' "=1 if obs has all the controls required for the models used"
										}
										
				foreach var in $r1_controls {
			replace has_all_controls_r1=has_all_controls_r1+1 if `var'==.
												}
												

********************************************************************************
*** Labelling ****
********************************************************************************
		forvalues x=1/2{
				label var kidsinsch715__`x' 		"Children aged 7-15 in school"
				label var n_children__`x' 			"Children aged below 15"
				label var sexratio715__`x' 			"Share of girls aged 7-15"		
				label var kidsinsch715__`x' 		"Children aged 7-15 in school"
				label var time_study_715_hh__`x' 	"Daily minutes studying for children aged 7-15"
				label var time_sch_715_hh__`x' 		"Daily minutes in school for children aged 7-15"
				label var kidsinhh715_F__`x' 		"Girls aged 7-15 in the household"
				label var inround`x'  				"=1 if in round `x'"
				label var A_4_2__`x'  				"Aspired Education (Years)"
				label var female_child__`x'			"=1 if child is female"
						}
						
				label var female 			"=1 if respondent is female"
				label var treatment 		"=1 if households is in treatment group"
				label var placebo 			"=1 if household is in placebo group"
				label var control 			"=1 if household is in control group"
				label var no_educ 			"=1 if respondent has no formal education"
				label var vil_num 			"Village ID"
				label var univ_id			"Respondent ID"
				label var hh_id				"Household (hh) ID"
				label var indivhhmerge__2	"=3 if respondent is the household head"

********************************************************************************
*** Removing unnecessary variables ****
********************************************************************************				
			drop new_control			///
				 sex_ID_child__1 		///
				 sex_ID_child__2 		///
				 above_med_init_regrade ///
				 indivhhmerge__1	
	 




DoFiles/PUBLICUSE_02_Analysis_Table_1_PanelA_v0.do
/*** I. Overview: File to produce the P&P results presented in Table 1 Panel A.
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com

****************************************
*** II. Version History ****
****************************************
2019/01/08, GS, created  from 03_Analysis_HTE_v3.do

****************************************
*** III. Inputs ****
****************************************
The summoning do-file:				"$dofiles\PUBLICUSE_00_Master_Analysis_v0.do"
The dataset: 						"$data/01_indivdatawide_PandP.dta"	- which will be turned into a household-level	
*/

********************************************************************************/	
	svyset hh_id
	
********************************************************************************
*** Define heterogeneity dimension ****
********************************************************************************
	global het 	female_child 	 

********************************************************************************
*** Run Regressions ****
********************************************************************************

foreach sub in $het 	{	
	global het `sub'
			
eststo clear
	foreach group in `ind_sr' {
	
		local nvars : list sizeof `group'
		local glogroup: list uniq `group'
		dis "`glogroup'"
		global glogroup `glogroup'
		dis "$glogroup"
		global gloname "`group'"
		dis "$gloname"
		display `nvars'
		matrix `group' = J(`nvars',11,.)
		local c = 0
		local d = 0
		
		foreach var in ``group''{
			
			local ++c
			
									
									local ++d
										 local vlabel_`d' :  variable label `var'__2		 
										* Spaces changed to _ 
										 local vlabel_`d' = subinstr("`vlabel_`d''", " ", "_", .)
										* Weird characters dropped as well
										 local vlabel_`d' = subinstr("`vlabel_`d''", "?", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "$", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "#", "", .)		 
										 local vlabel_`d' = subinstr("`vlabel_`d''", "@", "", .)

										 ** Creates and empty list to save the labels into a single line
										local labels_list = ""
										display "`labels_list'"
										** Number of variables. You need this to make the loop
										local numberofvars : list sizeof `group'
										display "`numberofvars'"
										** Loop to add all the corrected labels into a single line
										forvalues h=1/`numberofvars'{
										local labels_list   `labels_list'   `vlabel_`h''
										}
										* Check that thing
										display "`labels_list'"
										

		*main regression short run
					capture noisily {
						**COLUMN (1)**				ASPIRATIONS FOR EVERYONE   			/*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (1) and (2)  */
				sum `var'__1 									if female_child__1!=. & `cond1' 
								global mean_`var'_r1=`r(mean)'
								global sd_`var'_r1=`r(sd)'
								global N_`var'_r1=`r(N)'

						**COLUMN (2) **	Difference for Girls?
						
				areg `var'__1 i.female_child__1 $r1_controls 						  , absorb(vil_num) vce(cluster hh_id)  
								estimates store e_`var'_r1_mg			//stands for round1 margins
								global N_`var'_r1_mg=e(N)	

						**COLUMN (3)**	*main ATE regression short run 				/*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (3) and (4)  */	 
						
				areg `var'__2  placebo treatment $r2_controls  if female_child__2!=.  , absorb(vil_num) vce(cluster hh_id)
								estimates store e_`var'_ate_sr	
								global N_`var'_ate_sr=e(N)
							
						**COLUMN (4) ** *main HTE regression short run								
						
				areg `var'__2 above_med_init_`sub'__2 treatabove`sub'__2  placebo treatment $r2_controls , absorb(vil_num) vce(cluster hh_id)					
								estimates store e_`var'_hte_sr
								global N_`var'_hte_sr=e(N)


		}
		}
		
	*			do "$dofiles\PRIVATEUSE_03_Table_1_PanelA_v1.do"


	eststo clear
	}
	dis "bracket group closes"
	}
	dis "bracket het closes"


	
di "END of Table 1 Panel A"





DoFiles/PUBLICUSE_02_Analysis_Table_1_PanelB_v0.do
/*** I. Overview: File to produce the P&P results presented in Table 1 Panel B.
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com

****************************************
*** II. Version History ****
****************************************
2019/01/08, GS, created  from 03_Analysis_HTE_v3.do

****************************************
*** III. Inputs ****
****************************************
The summoning do-file:				"$dofiles\PUBLICUSE_00_Master_Analysis_v0.do"
The dataset: 						"$data/01_indivdatawide_PandP.dta"	- which will be turned into a household-level	
*/
	
	svyset hh_id
	
********************************************************************************
*** Define heterogeneity dimension and adapt controls to take out the extra dimension ****
********************************************************************************
	global het 	female_child#female
if "$het"=="female_child#female"{
					global r2_controls 		 age regrade_ID single E_1_6 E_1_7	hhsize__1 n_children__1
					global r1_controls 		 age regrade_ID single E_1_6 E_1_7	hhsize__1 n_children__1
}
********************************************************************************
*** Run Regressions ****
********************************************************************************

eststo clear
	foreach group in `ind_sr' {
	
		local nvars : list sizeof `group'
		local glogroup: list uniq `group'
		dis "`glogroup'"
		global glogroup `glogroup'
		dis "$glogroup"
		global gloname "`group'"
		dis "$gloname"
		display `nvars'
		matrix `group' = J(`nvars',11,.)
		local c = 0
		local d = 0
		
		foreach var in ``group''{
			
			local ++c
			
									
									local ++d
										 local vlabel_`d' :  variable label `var'__2		 
										* Spaces changed to _ 
										 local vlabel_`d' = subinstr("`vlabel_`d''", " ", "_", .)
										* Weird characters dropped as well
										 local vlabel_`d' = subinstr("`vlabel_`d''", "?", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "$", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "#", "", .)		 
										 local vlabel_`d' = subinstr("`vlabel_`d''", "@", "", .)

										 ** Creates and empty list to save the labels into a single line
										local labels_list = ""
										display "`labels_list'"
										** Number of variables. You need this to make the loop
										local numberofvars : list sizeof `group'
										display "`numberofvars'"
										** Loop to add all the corrected labels into a single line
										forvalues h=1/`numberofvars'{
										local labels_list   `labels_list'   `vlabel_`h''
										}
										* Check that thing
										display "`labels_list'"
										

					*main regression short run
					capture noisily {
							
							**OPTION COLUMN (1) **	ASPIRATIONS FOR WOMEN
																	
					areg `var'__1 i.female	$r1_controls /*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (1) and (2)  */	if female_child__1!=.   , absorb(vil_num) vce(cluster hh_id)
								estimates store e_`var'_r1					//stands for round1
								global N_`var'_r1=e(N)	
					

						**COLUMN (1) and (2) **	Difference for Girls?
						
					areg `var'__1 i.female##i.female_child__1 $r1_controls , absorb(vil_num) vce(cluster hh_id)
								estimates store e_`var'_r1_mg			//stands for round1 margins
								global N_`var'_r1_mg=e(N)	

							**OPTION COLUMN (3) **	ASPIRATIONS FOR WOMEN
																 
					areg `var'__2 i.female__2##i.treatment	placebo $r2_controls  /*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (3) and (4)  */ if female_child__2!=.  , absorb(vil_num) vce(cluster hh_id)
								estimates store e_`var'_hte1					
								global N_`var'_hte1=e(N)	
								
						**COLUMN (3) (4) ** *main HTE regression short run								

					areg `var'__2 i.female__2##i.treatment##i.female_child__2  placebo $r2_controls , absorb(vil_num) vce(cluster hh_id)
								estimates store e_`var'_hte_sr
								global N_`var'_hte_sr=e(N)

		}
		}
	*		do "$dofiles\PRIVATEUSE_03_Table_1_PanelB_v1.do"
		
	
	}
	dis "bracket group closes"

	
di "END of Table 1 Panel B"





DoFiles/PUBLICUSE_02_Analysis_Table_2_v0.do
/*** I. Overview: File to produce the P&P results presented in Table 2.
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com


****************************************
*** II. Version History ****
****************************************
2019/01/08, GS created from 03_Analysis_HTE_hh_v2.do

****************************************
*** III. Inputs ****
****************************************
The summoning do-file:				"$dofiles\PUBLICUSE_00_Master_Analysis_v0.do"
The dataset: 						"$data/01_indivdatawide_PandP.dta"	- which will be turned into a household-level	
*/

********************************************************************************
*** Define heterogeneity dimension and reconfirm controls****
********************************************************************************
	global het 	wm_sexratio${agegroup}  	 
	global r2_controls 				$ind_hh_controls  
	global r1_controls 				$r2_controls	
********************************************************************************
*** Turn dataset into Household-level ***
********************************************************************************
	keep if indivhhmerge__2==3		//This command keeps on observation per household, which refers to the household head
	
	drop indivhhmerge*				//Not necessary for analysis

	svyset hh_id
	
********************************************************************************
*** Run regressions ***
********************************************************************************

		foreach het_var in $het	{	

eststo clear
	foreach group in `hh_sr' {

	local nvars : list sizeof `group'
	local glogroup: list uniq `group'
	dis "`glogroup'"
	global glogroup `glogroup'
	dis "$glogroup"
	global gloname "`group'"
	dis "$gloname"
	display `nvars'
	matrix `group' = J(`nvars',11,.)
	local c = 0
	local d = 0
	
	
	foreach var in ``group''{
	
	
		
									 local ++c
									 local ++d
									 local vlabel_`d' :  variable label `var'__2		 
									* Spaces changed to _ 
									 local vlabel_`d' = subinstr("`vlabel_`d''", " ", "_", .)
									* Weird characters dropped as well
									 local vlabel_`d' = subinstr("`vlabel_`d''", "?", "", .)
									 local vlabel_`d' = subinstr("`vlabel_`d''", "$", "", .)
									 local vlabel_`d' = subinstr("`vlabel_`d''", "#", "", .)		 
									 local vlabel_`d' = subinstr("`vlabel_`d''", "@", "", .)

									 ** Creates and empty list to save the labels into a single line
									local labels_list = ""
									display "`labels_list'"
									** Number of variables. You need this to make the loop
									local numberofvars : list sizeof `group'
									display "`numberofvars'"
									** Loop to add all the corrected labels into a single line
									forvalues h=1/`numberofvars'{
									local labels_list   `labels_list'   `vlabel_`h''
									}
									* Check that thing
									display "`labels_list'"
							
		capture noisily {
						**COLUMN (1)**	BASELINE MEAN OUTCOMES FOR EVERYONE /*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (1) and (2)  */	 
			reg `var'__1 																	 	if `cond1', robust 		
								estimates store e_`var'_r1									//stands for round1
							    global N_`var'_r1=e(N)
						
						**COLUMN (2) **	Difference for Girls?
			areg `var'__1 `het_var'__1		1.nokidsinhh${agegroup}__1   $r1_controls 		 	, absorb(vil_num) robust														
								estimates store e_`var'_r1_mg								//stands for round1 margins
								global N_`var'_r1_mg=e(N)		

						**COLUMN (3)**	*main ATE regression short run /*  ADDING THE IF CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (3) and (4)  */
			areg `var'__2  treatment placebo $r2_controls 									 	 , absorb(vil_num) robust
								estimates store e_`var'_ate_sr	
								global N_`var'_ate_sr=e(N)		
						
						**COLUMN (4) ** *main HTE regression short run	- Difference for Girls?							
			areg `var'__2  treatment placebo c.`het_var'__2   1.treatment#c.`het_var'__2  	1.nokidsinhh${agegroup}__2  $r2_controls 		, absorb(vil_num) robust	//For continuous heterogeneity dimensions!
								estimates store e_`var'_hte_sr
								global N_`var'_hte_sr=e(N)		

							}							
							}							
	*							include "$dofiles/PRIVATEUSE_03_Table_2_v1"

				
							}

					eststo clear
							}
								di "closes het_var bracket"


di "END of Table 2"







DoFiles/PUBLICUSE_02_Analysis_Table_1_PanelC_v0.do
/*** I. Overview: File to produce the P&P results presented in Table 1 Panel C.
********************************************************************************
Author: Giulio Schinaia --- Email: w.giulioschinaia@gmail.com

****************************************
*** II. Version History ****
****************************************
2019/01/08, GS, created  from 03_Analysis_HTE_PanelC_v2.do

****************************************
*** III. Inputs ****
****************************************
The summoning do-file:				"$dofiles\PUBLICUSE_00_Master_Analysis_v0.do"
The dataset: 						"$data/01_indivdatawide_PandP.dta"	- which will be turned into a household-level	
*/

********************************************************************************/

	svyset hh_id
	
********************************************************************************
*** Define heterogeneity dimension and adapt controls to take out the extra dimension ****
********************************************************************************
global het 	female_child#noeduc		//this macro is just used to re-define the controlset, it doesn't go in the regressions
if "$het"=="female_child#noeduc"	{
global r2_controls 		 age male single E_1_6 E_1_7	hhsize__1 n_children__1
global r1_controls 		 age male single E_1_6 E_1_7	hhsize__1 n_children__1
}
********************************************************************************
*** Run Regressions ****
********************************************************************************
			
eststo clear
	foreach group in `ind_sr' {
	
		local nvars : list sizeof `group'
		local glogroup: list uniq `group'
		dis "`glogroup'"
		global glogroup `glogroup'
		dis "$glogroup"
		global gloname "`group'"
		dis "$gloname"
		display `nvars'
		matrix `group' = J(`nvars',11,.)
		local c = 0
		local d = 0
		
		foreach var in ``group''{
			
			local ++c
			
									
									local ++d
										 local vlabel_`d' :  variable label `var'__2		 
										* Spaces changed to _ 
										 local vlabel_`d' = subinstr("`vlabel_`d''", " ", "_", .)
										* Weird characters dropped as well
										 local vlabel_`d' = subinstr("`vlabel_`d''", "?", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "$", "", .)
										 local vlabel_`d' = subinstr("`vlabel_`d''", "#", "", .)		 
										 local vlabel_`d' = subinstr("`vlabel_`d''", "@", "", .)

										 ** Creates and empty list to save the labels into a single line
										local labels_list = ""
										display "`labels_list'"
										** Number of variables. You need this to make the loop
										local numberofvars : list sizeof `group'
										display "`numberofvars'"
										** Loop to add all the corrected labels into a single line
										forvalues h=1/`numberofvars'{
										local labels_list   `labels_list'   `vlabel_`h''
										}
										* Check that thing
										display "`labels_list'"
										

					*main regression short run
					capture noisily {
							
						**COLUMN (1) **	BASELINE NO_EDUC

					areg `var'__1 i.no_educ													 	$r1_controls /*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (1) and (2)  */	 if female_child__1!=.   , absorb(vil_num) vce(cluster hh_id)
							estimates store e_`var'_r1				//stands for round1
							global N_`var'_r1=e(N)	

						**COLUMN (2) **	BASELINE NO_EDUC INTERACTIONS

					areg `var'__1 i.no_educ 1.no_educ#1.female_child__1 1. female_child__1 		$r1_controls , absorb(vil_num) vce(cluster hh_id)
							estimates store e_`var'_r1_mg			//stands for round1 margins
							global N_`var'_r1_mg=e(N)	


						**COLUMN (3)** HTE NO_EDUC regression short run

					areg `var'__2 i.treatment##i.no_educ placebo 								 $r2_controls /*  ADDING THIS CONDITION TO HARMONIZE SAMPLE SIZE ACROSS COLUMN (3) and (4)  */	 if female_child__2!=. , absorb(vil_num) vce(cluster hh_id)
							estimates store e_`var'_hte1	
							global N_`var'_hte1=e(N)	

						**COLUMN (4) ** *main HTE NO_EDUCxFEMALE_CHILD regression short run								

					areg `var'__2 i.treatment##i.no_educ##i.female_child__2 placebo 			 $r2_controls , absorb(vil_num) vce(cluster hh_id)
							estimates store e_`var'_hte_sr
							global N_`var'_hte_sr=e(N)

		}
		}
		
	*			do "$dofiles\PRIVATEUSE_03_Table_1_PanelC_v1.do"

	
	}
	dis "bracket group closes"


	
di "END of Table 1 Panel C"

